```
<< NC`;
<< NCAlgebra`;
```

```
You are using the version of NCAlgebra which is found in:
  d:\Users\Johannes\Codes and Libraries\NC.
You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.
You have already loaded NCAlgebra.m
```

```
SetCommutative[h, μ];
(*this is needed, because of bug in Series*)series[f_, {x_, x0_, n_}] :=
 Sum[(x - x0)^k/k! Simplify[NCExpand[D[f, {x, k}] /. x → x0]], {k, 0, n}]
Pow[A_, n_] := Nest[(A ** # &), A, n - 1];
(*Calculates 1/(1+h A) as NC Series*)
Inv[A_, h_, n_] := 1 + Sum[(-h)^k Pow[A, k], {k, 1, n}]
exp[A_, h_, n_] := 1 + Sum[h^k Pow[A , k] / k!, {k, 1, n}]
T[hs_, As_] :=
 Inv[Through[As[t], Plus], -hs μ, 5] ** (1 + (1 - μ) hs Through[As[t + h], Plus])

r = D[u[t], t] → -A[t] ** u[t];
```

Test Standard Crank Nicolson

```
series[T[h, A] ** u[t + h] - u[t], {h, 0, 2}] /. D[r, t] /. r
```

$$\frac{1}{2} h^2 \left(-A'[t] ** u[t] - 2 (-1+\mu)\, A'[t] ** u[t] - A[t] ** A[t] ** u[t] + 2 \mu\, A[t] ** A[t] ** u[t]\right)$$

```
% /. μ → 1 / 2
```

0

Basic Splitting

```
n = Range[1];
series[Fold[#1 ** #2 &, T[h, A + Total[B[#] & /@ n]], T[-h, B[#]] & /@ n] ** u[t + h] - u[t],
   {h, 0, 2}] /. D[r, t] /. r
```

$$\frac{1}{2} h^2 \big(-A'[t] ** u[t] - 2 (-1+\mu)\, A'[t] ** u[t] - A[t] ** A[t] ** u[t] +$$
$$2 \mu\, A[t] ** A[t] ** u[t] - 2\, A[t] ** B[1][t] ** u[t] + 2 \mu\, A[t] ** B[1][t] ** u[t] +$$
$$2 \mu\, B[1][t] ** A[t] ** u[t] - 2\, B[1][t] ** B[1][t] ** u[t] + 4 \mu\, B[1][t] ** B[1][t] ** u[t]\big)$$

```
% /. μ → 1 / 2
```

$$\frac{1}{2} h^2 (-A[t] ** B[1][t] ** u[t] + B[1][t] ** A[t] ** u[t])$$

Strang Symmetrized Splitting

```
n = Range[4];
op = Fold[#2 ** #1 &, Fold[#1 ** #2 &, G[h, A + Total[B[#] & /@ n]], G[-h / 2, B[#]] & /@ n],
  G[-h / 2, B[#]] & /@ n]
```

$$G\left[-\frac{h}{2}, B[4]\right] ** G\left[-\frac{h}{2}, B[3]\right] ** G\left[-\frac{h}{2}, B[2]\right] ** G\left[-\frac{h}{2}, B[1]\right] ** G[h, A + B[1] + B[2] + B[3] + B[4]] ** G\left[-\frac{h}{2}, B[1]\right] ** G\left[-\frac{h}{2}, B[2]\right] ** G\left[-\frac{h}{2}, B[3]\right] ** G\left[-\frac{h}{2}, B[4]\right]$$

```
Simplify[NCExpand[series[(op /. G → T) ** u[t + h] - u[t], {h, 0, 2}] /. D[r, t] /. r]]
```

$$-\frac{1}{4} h^2 (-1 + 2 \mu) (2 A'[t] ** u[t] - 2 A[t] ** A[t] ** u[t] - 2 A[t] ** B[1][t] ** u[t] - 2 A[t] ** B[2][t] ** u[t] - 2 A[t] ** B[3][t] ** u[t] - 2 A[t] ** B[4][t] ** u[t] - 2 B[1][t] ** A[t] ** u[t] - 3 B[1][t] ** B[1][t] ** u[t] - 2 B[1][t] ** B[2][t] ** u[t] - 2 B[1][t] ** B[3][t] ** u[t] - 2 B[1][t] ** B[4][t] ** u[t] - 2 B[2][t] ** A[t] ** u[t] - 2 B[2][t] ** B[1][t] ** u[t] - 3 B[2][t] ** B[2][t] ** u[t] - 2 B[2][t] ** B[3][t] ** u[t] - 2 B[2][t] ** B[4][t] ** u[t] - 2 B[3][t] ** A[t] ** u[t] - 2 B[3][t] ** B[1][t] ** u[t] - 2 B[3][t] ** B[2][t] ** u[t] - 3 B[3][t] ** B[3][t] ** u[t] - 2 B[3][t] ** B[4][t] ** u[t] - 2 B[4][t] ** A[t] ** u[t] - 2 B[4][t] ** B[1][t] ** u[t] - 2 B[4][t] ** B[2][t] ** u[t] - 2 B[4][t] ** B[3][t] ** u[t] - 3 B[4][t] ** B[4][t] ** u[t])$$

```
Simplify[% /. μ → 1 / 2]
```

$$0$$

Lifting the inhomogenity

```
r2 = D[u[t], t] → -A ** u[t] - b[t];
```

```
CN[h_, μ_] := Inv[A, -h μ, 5] ** ((1 + h A (1 - μ)) ** # + h (μ b[t] + (1 - μ) b[h + t])) &
CN2[h_, μ_] := Inv[A, -h μ, 5] ** ((1 + h A (1 - μ)) ** # + h ((1 - μ) b[h + t])) + h μ b[t] &
```

```
Simplify[NCExpand[series[CN[h, μ][u[t + h]] - u[t], {h, 0, 3}] /. D[r2, t] /. r2]]
```

$$\frac{1}{6} h^2 \left((-3 + 6 \mu) A ** b[t] - 3 h \left(1 - 2 \mu + 2 \mu^2\right) A ** b'[t] + 3 h A ** A ** b[t] - 6 h \mu A ** A ** b[t] + 6 h \mu^2 A ** A ** b[t] - 3 A ** A ** u[t] + 6 \mu A ** A ** u[t] + 3 h A ** A ** A ** u[t] - 6 h \mu A ** A ** A ** u[t] + 6 h \mu^2 A ** A ** A ** u[t] + 3 b'[t] - 6 \mu b'[t] + 3 h b''[t] - 3 h \mu b''[t] + h u^{(3)}[t]\right)$$

```
Simplify[NCExpand[% /. μ → 1 / 2]]
```

$$\frac{1}{12} h^3 \left(-3 A ** b'[t] + 3 A ** A ** b[t] + 3 A ** A ** A ** u[t] + 3 b''[t] + 2 u^{(3)}[t]\right)$$

```
Simplify[NCExpand[series[CN2[h, μ][u[t + h]] - u[t], {h, 0, 3}] /. D[r2, t] /. r2]]
```

$$\frac{1}{6} h^2 \left(\left(-3+6\mu-6\mu^2\right) A ** b[t] - 3 h \left(1-2\mu+2\mu^2\right) A ** b'[t] + 3 h A ** A ** b[t] - 6 h \mu A ** A ** b[t] + 6 h \mu^2 A ** A ** b[t] - 6 h \mu^3 A ** A ** b[t] - 3 A ** A ** u[t] + 6 \mu A ** A ** u[t] + 3 h A ** A ** A ** u[t] - 6 h \mu A ** A ** A ** u[t] + 6 h \mu^2 A ** A ** A ** u[t] + 3 b'[t] - 6 \mu b'[t] + 3 h b''[t] - 3 h \mu b''[t] + h u^{(3)}[t]\right)$$

```
Simplify[NCExpand[% /. μ → 1 / 2]]
```

$$\frac{1}{24} h^2 \left(-6 A ** b[t] + h \left(-6 A ** b'[t] + 3 A ** A ** b[t] + 6 A ** A ** A ** u[t] + 6 b''[t] + 4 u^{(3)}[t]\right)\right)$$

```
CN[h, μ]
```

Inv[A, -h μ, 5] ** ((1 + h A (1 - μ)) ** #1 + h (μ b[t] + (1 - μ) b[h + t])) &

```
CN2[h, μ]
```

Inv[A, -h μ, 5] ** ((1 + h A (1 - μ)) ** #1 + h ((1 - μ) b[h + t])) + h μ b[t] &